# $Id: PKGBUILD 184164 2013-05-03 01:57:18Z heftig $
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: Sarah Hay <sarahhay@mb.sympatico.ca>

# This package must not by "any" arch.
# Requires rebuild with every change to sdl-config

pkgname=perl-alien-sdl
_realname=Alien-SDL
pkgver=1.440
pkgrel=1
pkgdesc="Building, finding and using SDL binaries"
arch=(i686 x86_64)
license=(PerlArtistic GPL)
url="http://search.cpan.org/dist/Alien-SDL/"
depends=(sdl=1.2.15 perl-file-sharedir perl-capture-tiny)
makedepends=(perl-file-which perl-text-patch)
options=(!emptydirs)
_author=FROGGS
source=(http://search.cpan.org/CPAN/authors/id/${_author::1}/${_author::2}/$_author/$_realname-$pkgver.tar.gz)
md5sums=('ad665d058d186e673114024ba10dfa09')

package() {
  cd "$srcdir/$_realname-$pkgver"

  # install module in vendor directories
  perl Build.PL installdirs=vendor destdir="$pkgdir" --with-sdl-config

  perl Build
  perl Build install
}
